

プログラムのページ

担当 森 口 繁 一

6301. 連立常微分方程式の解法 (Runge-Kutta-Gill 法)

森口繁一 (東大工学部)

微分方程式

$$\frac{dy_i}{dx} = f_i(x, y_1, \dots, y_m) \quad (i=1, 2, \dots, m)$$

を, 初期条件

$$y_i(a) = \eta_i \quad (i=1, \dots, m)$$

の下で解くものとする.

x を h キザミで増す. ある x に対する y_i の値を y_{i0} として, 次の計算で求めた y_{i4} を $x+h$ に対する y_i の値とする.

$$\begin{aligned} k_{i0} &= hf_i(y_{00}, y_{10}, \dots, y_{n0}), \\ r_{i1} &= (1/2)(k_{i0} - 2q_{i0}), \\ y_{i1} &= y_{i0} + r_{i1}, \\ q_{i1} &= q_{i0} + 3r_{i1} - (1/2)k_{i0}, \\ k_{i1} &= hf_i(y_{01}, y_{11}, \dots, y_{n1}), \\ r_{i2} &= (1 - \sqrt{1/2})(k_{i1} - q_{i1}), \\ y_{i2} &= y_{i1} + r_{i2}, \\ q_{i2} &= q_{i1} + 3r_{i2} - (1 - \sqrt{1/2})k_{i1}, \\ k_{i2} &= hf_i(y_{02}, y_{12}, \dots, y_{n2}), \\ r_{i3} &= (1 + \sqrt{1/2})(k_{i2} - q_{i2}), \\ y_{i3} &= y_{i2} + r_{i3}, \\ q_{i3} &= q_{i2} + 3r_{i3} - (1 + \sqrt{1/2})k_{i2}, \\ k_{i3} &= hf_i(y_{03}, y_{13}, \dots, y_{n3}), \\ r_{i4} &= (1/6)(k_{i3} - 2q_{i3}), \\ y_{i4} &= y_{i3} + r_{i4}, \\ q_{i4} &= q_{i3} + 3r_{i4} - (1/2)k_{i3}, \\ &\quad (i=0, 1, \dots, m). \end{aligned}$$

ただし, 式中, x を y_0 で表わしてあり, したがって, $f_0(y_0, y_1, \dots, y_m)$ の値はつねに 1 に等しく, y_{00} は x 自身である. q_{i0} は出発点ではすべて 0 とするが, あとは一つ前のステップでの q_{i4} を q_{i0} とする.

プログラム

RKGRENTRITU:

```
begin real H, K, R, QI;
integer N, NN, I, M;
array Y, Q, F [0: 10];
procedure UHEN;
```

begin F [1] := Y[2];

F [2] := Y[0] × Y[2] + Y[1]

end;

procedure INSATU;

begin CRLF; PRINTSTRING('X=');

PRINTREAL(Y[0]);

for I:=1 step 1 until M do

begin CRLF; PRINTREAL(Y[I]);

PRINTREAL(F[I])

end

end;

HAZIME: READREAL(H); READINTEGER(NN);

CRLF; PRINTSTRING('H='); PRINTREAL(H); CRLF;

READINTEGER(M);

for I:=0 step 1 until M do

begin READREAL(Y[I]); Q[I]:=0 end;

F[0]:=1.0;

UHEN; INSATU;

SINKO: for N:=1 step 1 until NN do

begin for I:=0 step 1 until M do

begin K:=H×F[I]; QI:=Q[I];

R:=0.5×K-QI;

Q[I]:=QI+3.0×R-0.5×K;

Y[I]:=Y[I]+R

end;

UHEN;

for I:=0 step 1 until M do

begin K:=H×F[I]; QI:=Q[I];

R:=0.29289322×(K-QI);

Q[I]:=QI+3.0×R-0.29289322×K;

Y[I]:=Y[I]+R

end;

UHEN;

for I:=0 step 1 until M do

begin K:=H×F[I]; QI:=Q[I];

R:=1.70710678×(K-QI);

Q[I]:=QI+3.0×R-1.70710678×K;

Y[I]:=Y[I]+R

```

end;
UHEN;
for I:=0 step 1 until M do
begin K:=H×F[I]; QI:=Q[I];
R:=(K-2.0×QI)/6.0;
Q[I]:=QI+3.0×R-0.5×K;
Y[I]:=Y[I]+R
end;
UHEN;
INSATU
end

```

〔注 1〕 H には h を入れる。K には $h f_i$ を、R には r_{ij} を、QI には q_{ij} を (一時的に)、それぞれ入れる。N でステップ数を数え、NN にはその最終値を入れておく。I には i 、M には m を入れる。Y[I] には y_{ij} 、Q[I] には q_{ij} 、F[I] には $f_i(y_0, y_1, \dots, y_m)$ を入れる。

〔注 2〕 UHEN (右辺) という名の手続は $y_0, y_1, \dots, y_m$ に対する f_i の値を計算するものである。ここでは具体例として $f_i(y_0, y_1, y_2) = y_2$ 、 $f_2(y_0, y_1, y_2) = y_0 y_2 + y_1$ の場合が示してある。INSATU (印刷) という名の手続は、まず x を第 1 行に、それから y_i, f_i ($i = 1, \dots, m$) をならべて m 行にわたって印刷するものである。

〔注 3〕 入力データとしては、キザミ h 、全ステップ数 NN、従属変数の数 m 、初期値 $a, \eta_1, \dots, \eta_m$ を、この順序に与える。

〔注 4〕 プログラム中の定数の由来

$$\begin{aligned} 0.29289322 &= 1 - \sqrt{1/2}, \\ 1.70710678 &= 1 + \sqrt{1/2}. \end{aligned}$$

6302. 連立常微分方程式の解法 (Milne 法)

森口繁一 (東大工学部)

微分方程式

$$\frac{dy_i}{dx} = f_i(x, y_1, \dots, y_m) \quad (i=1, 2, \dots, m)$$

を、初期条件

$$y_i(a) = \eta_i \quad (i=1, \dots, m)$$

の下で解くものとする。

x を h キザミで増し、 $x_n = a + nh$ とする。 $x = x_n$ に対する y_i の値を $y_{i,n}$ 、そして、そこでの右辺の値を $f_{i,n}$ と書く。予測の公式は

$$y_{i,n+4}^p = y_{i,n} + \frac{4}{3} h (2f_{i,n+1} - f_{i,n+2} + 2f_{i,n+3}),$$

修正の公式は

$$y_{i,n+3}^c = y_{i,n+1} + \frac{h}{3} (f_{i,n+1} + 4f_{i,n+2} + f_{i,n+3})$$

である (ただし、予測のあと番号 n を 1 だけ進めてから修正の公式を使う)。第 1 回の修正量 C_{0i} は精度の目安になるので記録しておく。修正量が、あらかじめ指定した定数 ϵ よりも小さくなったら収束したものととして次に進む。

プログラム

MILNERENRITU:

```

begin real H, EPS, X, YP, YC, C;
integer N, NN, I, M, NC, S1, S2;
array Y0, Y1, Y2, F1, F2, Y, F, C0[1:10];
procedure UHEN;
begin F[1]:=Y[2];
F[2]:=X×Y[2]+Y[1]
end;
procedure INSATU;
begin CRLF; PRINTSTRING('X=');
PRINTREAL(X);
PRINTSTRING('NC=');
PRINTINTEGER(NC);
for I:=1 step 1 until M do
begin CRLF; PRINTREAL(Y[I]);
PRINTREAL(F[I]); PRINTREAL(C0[I])
end
end;

```

procedure SYUPPATUTI;

```

begin for I:=1 step 1 until M do
READREAL(Y[I]);
UHEN; INSATU
end;

```

```

HAZIME: READREAL(EPS); READREAL(H);
CRLF; PRINTREAL(EPS);
PRINTREAL(H); CRLF;
READINTEGER(NN); READINTEGER(M);
NC:=0; for I:=1 step 1 until M do
C0[L]:=0.0;
READREAL(X); SYUPPATUTI;
for I:=1 step 1 until M do Y0[I]:=Y[I];
X:=X+H; SYUPPATUTI;
for I:=1 step 1 until M do
begin Y1[I]:=Y[I]; F1[I]:=F[I] end;
X:=X+H; SYUPPATUTI;
for I:=1 step 1 until M do
begin Y2[I]:=Y[I]; F2[I]:=F[I] end;
X:=X+H; SYUPPATUTI;
SINKO: for N:=4 step 1 until NN do
begin X:=X+H; NC:=0; S1:=0;
YOSOKU: for I:=1 step 1 until M do
begin YP:=Y0[I]+(2.0×F1[I]-F2[I]
+2.0×F[I])×H/0.75;
ZENSIN: Y0[I]:=Y1[I]; Y1[I]:=Y2[I];

```